
Informatici

1. Subroutine con valori di ritorno in locazioni di memoria

2. Subroutine con parametri e valori di ritorno in registri

4. Minimo tra due valori mediante lo stack e allocazione dinamica

5. Minimo fra 3 valori mediante stack e allocazione dinamica

6. Calcolo della media degli elementi di due vettori v1 e v2

7. Esempio di passaggio di parametri per indirizzo

8. Esempio di passaggio per valore

9. Lettura dati da periferica e calcolo della somma

10. Lettura dati da periferica, calcolo della somma e invio di essa ad altra periferica

11. Invio ad una periferica di una sequenza di valori memorizzati in un vettore e terminata con uno 0.

12. Acquisizione di dati da una periferica e loro trasferimento in memoria a partire da un dato indirizzo.

13. Come l’esercizio precedente, ma strutturando il programma con una subroutine che riceva come parametri il numero dei dati da acquisire, l’indirizzo di memoria in cui salvare i dati e l’indirizzo del dispositivo

14. Calcolo, mediante due subroutine distinte, della somma degli elementi pari e di quella dei dispari di un vettore v e invio delle somme ad una periferica out.

15. Programma con routine generalizzata che consenta il polling a ciclo continuo su 3 periferiche di output (non utilizzando l’IVN del PD32)

16. Programma di invio ad una periferica out, di indirizzo 30h, di un vettore di n elementi word. L’invio dei dati in output deve avvenire al momento della ricezione di un segnale proveniente da una periferica di indirizzo 20h.
17. Acquisire da input una serie di N valori da una periferica di indirizzo 12h e porli in un vettore con indirizzo base 400h. Utilizzare il sistema di interruzione.

18. Acquisire N valori long word da un dispositivo di input, sincronizzando l’immissione di ciascun dato con un segnale di clock proveniente da un dispositivo esterno. Al termine dell’immissione, inviare i dati acquisiti ad una periferica di output.

Esercizio 1

Un primo modo di eseguire le subroutine è quello di utilizzare una data locazione di memoria per contenere i valori di ritorno. Nel seguente esempio, una subroutine addiz e una sottr ricevono i parametri dai registri R0 ed R1, elaborano i valori numerici e restituiscono il risultato in una locazione di memoria opportuna.
;Esempio di subroutine con parametri in registri e valori 

;di ritorno in locazione di memoria

;Autore: M.M.

;Data: 10/04/2004

;File: sbrlocaz.asm

   ORG 400h

   somma   dl 0

   diff    dl 0

CODE

   MOVL #3, R0     ;preparazione parametri

   MOVL #5, R1

   JSR  addiz

   MOVL #5, R0     ;preparazione parametri

   MOVL #-2,R1  

   JSR  sottr

   JMP  fine

;**** Subroutine *************************************

addiz:

   MOVL R0, R3

   ADDL R1, R3

   MOVL R3           , somma  ;valore di ritorno

   RET

sottr:

   MOVL R0, R3

   SUBL R1, R3

   MOVL R3, diff

   RET

******************************************************

fine:

   HALT

END

Esercizio 2

Un altro modo di eseguire le subroutine è quello di utilizzare i registri sia per passare i parametri di ingresso, sia per contenere i valori di ritorno. Nel seguente esempio, la subroutine somma riceve i parametri in R0 ed R1, esegue la somma e restituisce il risultato in R3
;Esempio di subroutine con parametri e valori di 
;ritorno in registri

;Autore: M.M.

;Data: 10/04/2004

;File: sbrreg.asm

ORG 400h

   ris1 dl 0

   ris2 dl 0

CODE

   MOVL #3, R0     ;preparazione parametri

   MOVL #5, R1

   JSR  somma

   MOVL R3, ris1   ;salvataggio valore di ritorno

   MOVL #2, R0     ;preparazione parametri

   MOVL #4, R1

   JSR  somma

   MOVL R3, ris2   ;salvataggio valore di ritorno

   JMP  fine

somma:

   MOVL #0, R3

   ADDL R0, R3

   ADDL R1, R3  ;valore di ritorno

   RET

fine:

   HALT

END

Esercizio 3

I parametri di una subroutine possono essere passati anche mediante l’uso dello stack. In questo caso, occorre salvare il frame precedente e, quindi, insieme allo Stack Pointer (registro R7) è necessario usare un altro registro (tipicamente R6) come Frame Pointer. In questo esempio, si realizza lo scambio dei valori di due registri mediante una subroutine che fa uso dello stack. Ovviamente si tratta di un esempio didattico, in quanto lo stesso problema si poteva risolvere in modo elementare, ma serve a chiarire lo schema dell’allocazione dinamica e della successiva deallocazione. Si osservi che alla fine del programma, nello stack sono ancora presenti dei valori, ma il registro R7 è tornato al valore iniziale (27FC) che corrisponde allo stack vuoto. Eventuali valori immessi successivamente nello stack, andranno a sovrascrivere i valori fisicamente presenti. 
;Programma di scambio di due valori con subroutine e uso 

;dello stack

;Autore: M.M.

;Data: 20/04/2004

;File: sbrstack.asm

ORG 400h

   a DL 5

   b DL 3

CODE

   MOVL a, R0

   MOVL b, R1

   PUSH R0         ;inserisce dati nello stack       

   PUSH R1

   JSR Scambia

   ADDL #8, R7     ;ripristina R7

   JMP Fine

Scambia:

   PUSH R6         ;salva frame precedente

   MOVL R7, R6     ;frame attuale

   PUSH R0         ;salva registri

   PUSH R1

   MOVL 8(R6), -4(R6) ;scambia i valori

   MOVL 12(R6), -8(R6)

   POP R1          ;ripristina i registri   

   POP R0

   POP R6          ;ripristina il frame precedente

   RET

Fine:

   HALT 

END

Esercizio 4
     In questo esempio si calcola il minimo tra due valori mediante una subroutine che fa uso dell’allocazione dinamica dello stack. Anche in questo caso si osservi il valore di R7 alla fine del programma. Gli elementi ancora fisicamente presenti nello stack, non sono logicamente presenti in quanto R7 punta alla base (27FC) che corrisponde allo stack logicamente vuoto (ma non necessariamente fisicamente)
;Programma per il calcolo del minimo 

;fra 2 numeri mediante lo stack

;Autore: M.M.

;Data: 10/04/2004

;File: sbrstck2.asm

ORG 400h

   a   DL 8

   b   DL 5

   vmin DL 0

CODE

   MOVL a, R0

   MOVL b, R1

   PUSH R0

   PUSH R1

   JSR min

   ADDL #8, R7

   JMP fine

min:

   PUSH R6      ;salva frame precedente

   MOVL R7, R6  ;frame attuale

   PUSH R2      ;salva registro usato

   CMPL R0, R1  ;calcola il minimo

   JNN minR0

   MOVL R1, R2

   JMP finemin

minR0:

   MOVL R0, R2

finemin:

   MOVL R2, vmin

   POP R2       ;ripristina R2

   POP R6       ;ripristina R6  

   RET

fine:

   HALT

END

Esercizio 5

Questo esercizio, simile al precedente, calcola il minimo fra 3 valori mediante stack e allocazione dinamica. Lo schema è sempre lo stesso: salvataggio del frame precedente, allocazione del frame attuale, elaborazione dei parametri e rilascio del frame correntemente allocato.
;Programma per il calcolo del minimo 

;fra 3 numeri mediante lo stack

;Autore: M.M.

;Data: 10/04/2004

;File: sbrstck3.asm

ORG 400h

   a   DL 8

   b   DL 2

   c   DL 6 

   vmin DL 0
CODE

   MOVL a, R0   ;parametri

   MOVL b, R1

   PUSH R0

   PUSH R1

   JSR min

   MOVL c, R0   ;parametri

   MOVL vmin, R1

   JSR min

   ADDL #8, R7  ;aggiorna R7

   JMP fine

min:

   PUSH R6      ;salva frame precedente

   MOVL R7, R6  ;frame attuale

   PUSH R2

   CMPL R0, R1

   JNN minR0

   MOVL R1, R2

   JMP finemin

minR0:

   MOVL R0, R2

finemin:

   MOVL R2, vmin

   POP R2       ;ripristina R2

   POP R6

   RET

fine:

   HALT

END

Esercizio 6


Dati due vettori di word, memorizzati a partire dalla locazione 400h, calcolare la media intera dei valori in essi contenuti. Utilizzare un’unica subroutine somma per il calcolo della somma e un’unica subroutine media per il calcolo della media

;Programma che calcola la media dei valori di due vettori

;mediante chiamata a subroutine

;Autore: M.M.

;Data:10/02/2004

;File: sbrvet1.asm

Org 400h


V1 DW 2,-2,8,4  ;caricamento dati


V2 DW 5,6,-4,9


SOMMA1 DW 0     ;inizializzazione variabili


SOMMA2 DW 0


MEDIA1 DW 0


MEDIA2 DW 0


N DW 4

CODE

;****** elaborazione vettore V1 ***************


MOVL #V1, R3    ;R3 parametro input


JSR somma


MOVW R2, SOMMA1 ;R2 parametro output


MOVW R2, R3     ;R3 parametro input


JSR media


MOVW R1, MEDIA1 ;R1 parametro output

;****** elaborazione vettore V2 ***************


MOVL #V2, R3    ;R3 parametro input


JSR somma


MOVW R2, SOMMA2 ;R2 parametro output


MOVW R2, R3
;R3 parametro input


JSR media


MOVW R1, MEDIA2 ;R1 parametro output


JMP fine

;**********************************************

;*      SUBROUTINE                            *

;**********************************************

;

;****** somma elementi vettore ****************

;* input : R3 indirizzo base vettore          *

;* output: R2 somma elementi                  *

;**********************************************

somma:
MOVW #0, R2   
;R3 indirizzo vettore


MOVB #0, R1

ciclo:
CMPW N, R1


JZ finecicl


ADDW (R3)+, R2


ADDB #1, R1


JMP ciclo

finecicl: RET

;

;****** calcolo media *************************

;* input : R3 valore somma                    *

;* output: R1 media elementi                  *

;**********************************************

media:  MOVB #0,R1

ciclom:
CMPW N, R3


JN endm


SUBW N, R3


ADDW #1, R1


JMP ciclom

endm:   RET

fine:
HALT

END

Esercizio 7

In questo esempio si mostra come sia possibile passare per indirizzo un parametro ad una subroutine. Il valore iniziale di op1 è messo in R2. Dopo la chiamata alla subroutine, che riceve l’indirizzo di op1 in R1, viene calcolato il doppio e posto come contenuto dell’indirizzo R1. Il nuovo valore di op1 è posto in R4. 
;Esempio di subroutine con parametro

;passato per indirizzo

;Al momento dell'attivazione si passa come parametro

;l'indirizzo dell'operando.

;Mediante indirizzamento indiretto si opera su esso 

;nella subroutine e il valore di op1 viene visto 

;modificato anche dal programma chiamante. 

;Autore: M.M.

;Data: 10/04/2004

;File: sbrxindi.asm

   ORG 400h

   op1 DL 5

CODE

   MOVL op1, R2   ;valore iniziale di op1

   MOVL #op1, R1  ;op1 passato per indirizzo

   JSR  doppio    ;risultato in op1

   MOVL op1, R4   ;valore finale di op1 (modificato)

   JMP  fine

doppio:           

   MOVL (R1), R3  ;variabile locale

   ASLL #1, R3  

   MOVL R3, (R1)   

   RET

fine:

   HALT

END

Esercizio 8

Un parametro passato per valore ad una subroutine non viene modificato al ritorno al programma chiamante. In questo esempio il parametro op1 è passato per valore alla subroutine mediante il registro R1. La subroutine ne calcola il doppio e torna al programma principale. Qua il valore di op1 non risulta modificato, in quanto la subroutine ha lavorato su una copia di op1.

;Esempio di subroutine con passaggio per valore

;Il parametro op1 viene copiato in R1. L'esecuzione della

;subroutine modifica R1 (side effect) ma non op1.

;Al ritorno al chiamante op1 resta invariato. 

;Autore: M.M.

;Data: 10/04/2004

;File: sbrxval.asm

   ORG 400h

   op1 DL 5

CODE

   MOVL op1, R1  ;R1 parametro formale

   JSR  doppio   

   MOVL op1, R2 

   JMP  fine

doppio:

   ASLL #1, R1   ;valore locale 

   RET

fine:

   HALT

END

Esercizio 9

Scrivere un programma in PD32 che legge da una periferica dev, posta all’indirizzo 34h, 5 dati di tipo byte, ne calcola la somma e la memorizza in una variabile somma.

Soluzione

Analisi

Il problema si risolve definendo la periferica dev con il suo indirizzo, e le variabili dati e somma di tipo DL.


Il programma azzera i registri da usare, esegue un ciclo di lettura dei dati dalla periferica ed accumula il risultato nella variabile somma.


Per la lettura dei dati dalla periferica si usa la tecnica busy waiting, con cui la CPU testa ripetutamente lo stato della periferica fino a che non risulti pronta e disponibile al trasferimento. Quando si verifica questa eventualità, il protocollo bosy waiting fa sì che la CPU attenda la produzione del dato dalla periferica e solo quando questo è stato completamente trasferito nel data buffer, può essere salvato in un registro di appoggio per essere trattato.

Registri usati:

R0: buffer per lettura dato

R1: accumulatore per la somma

R2: contatore dati

Algoritmo
Inizio

   Azzeramento registri;

   Lettura dati;

Fine.

Lettura dati

Inizio

   Per  R2 ( 5 a 0 fai
   Inizio

        Fintantochè dispositivo non pronto attendi;
        Avvia dispositivo;
        Fintantochè dato non trasferito attendi;

        R0 ( data buffer;

        R1 ( R0 + R1;
   Fine;

   Somma ( R1;
Fine.

Codifica

;Programma che legge 5 dati di tipo byte dalla periferica

;p1 e ne calcola la somma

;Autore: M.M.

;Data: 10/04/2004

;File: input1.asm

ORG 400h

    dev   EQU 34h      ;indirizzo disp. Input

    somma DL 0         ;risultato

    dati  DL 5              ;numero dati da leggere

CODE

    JSR initregs         ;azzera registri

    JSR dataread       ;lettura dati

    JMP exit

;****** Inizializza registri usati ********

initregs:

    XORL R0, R0        ;buffer lettura

    XORL R1, R1        ;accumulatore per somma

    XORL R2, R2        ;contatore dati  

    RET

;****** Acquisizione dato **********************

datain:

devnotrd:

    JNR dev, devnotrd  ;attesa che il dispositivo si



            ;si renda libero

    START dev             ;avviamento produzione del dato

datnotrd: 

    JNR dev, datnotrd  ;attesa che il dato sia disponibile

    INB dev, R0           ;leggo il dato

    RET

;****** Lettura dati **********************

dataread:

    MOVL dati, R2

ciclo:

    JSR  datain              ;legge singolo dato

    ADDL R0, R1

    SUBL #1, R2

    JNZ ciclo

    MOVL r1, somma

    RET

;*******************************************

exit:

    HALT

END

Esercizio 10
Scrivere un programma PD32 che acquisisca dalla periferica inp, posta ad indirizzo 34h, una sequenza di 10 dati di tipo byte, ne calcoli la somma e invii detta somma alla periferica out, posta all’indirizzo 10h.

Soluzione
Analisi

Il problema si risolve definendo le due periferiche inp e out, con i rispettivi indirizzi, e le variabili dati e somma di tipo DL.


Il programma azzera i registri da usare, esegue un ciclo di lettura dei dati dalla periferica inp ed accumula il risultato nella variabile somma.

Successivamente la variabile somma è messa in output sulla periferica out.

Sia per la lettura dei dati dalla periferica, che per l’invio della somma alla periferica out, si usa la tecnica busy waiting, in cui la CPU attende lo stato di “pronto” delle perifericha rispettive. La periferica inp produce i singoli dati, la periferica out consuma il dato prodotto dal calcolo della somma.

Registri usati:

R0: buffer per lettura dato

R1: accumulatore per la somma

R2: contatore dati

Algoritmo
Inizio

   Azzeramento registri;

   Lettura dati;
   Stampa risultato;
Fine.

Lettura dati

Inizio

   Per  R2 ( 5 a 0 fai

   Inizio

        Fintantochè dispositivo non pronto attendi;

        Avvia dispositivo;

        Fintantochè dato non trasferito attendi;

        R0 ( data buffer;

        R1 ( R0 + R1;

   Fine;

   Somma ( R1;

Fine.

Stampa risultato
Inizio

         Fintantochè dispositivo non pronto attendi;
         data buffer ( somma;

         Avvia dispositivo;

Fine;        
;Programma che legge 10 dati di tipo byte dalla periferica

;p1, ne calcola la somma e la invia alla periferica p2

;Autore: M.M.

;Data: 10/04/2004

;File: inout1.asm

ORG 400h

    inp   EQU 34h      ;indirizzo disp. Input

    out   EQU 10h      ;indirizzo disp. output

    somma DL 0         ;risultato

    dati  DL 10            ;numero dati da leggere

CODE

    JSR initregs           ;azzera registri

    JSR dataread          ;lettura dati

    JSR dataout

    JMP exit

;****** Inizializza registri usati ********

initregs:

    XORL R0, R0        ;buffer lettura

    XORL R1, R1        ;accumulatore per somma

    XORL R2, R2        ;contatore dati  

    RET

;****** Acquisizione dato **********************

datain:

inpnotrd:

    JNR inp, inpnotrd  ;attesa che il dispositivo si



           ;si renda libero

    START inp            ;avviamento produzione del dato

datnotrd: 

    JNR inp, datnotrd  ;attesa che il dato sia disponibile

    INB inp, R0           ;leggo il dato

    RET

;****** Lettura dati **********************

dataread:

    MOVL dati, R2

ciclo:

    JSR  datain              ;legge singolo dato

    ADDL R0, R1

    SUBL #1, R2

    JNZ ciclo

    MOVL r1, somma

    RET

;****** Emissione risultato ***************

dataout:

outnotrd:

    JNR out, outnotrd    ;attesa fino a disponibilità di out
    OUTL somma, out  ;caricamento data buffer di out
    START out             ;avvio del consumo del dato
    RET

;******************************************

exit:

    HALT

END

Esercizio 11
Scrivere un programma in PD32 che, dato un vettore v, posto all’indirizzo 400h, contenente una sequenza di valori di tipo long, terminata con uno 0, ne invii il contenuto, elemento per elemento, ad una periferica out, posta all’indirizzo 10h.

Soluzione

Analisi

Il problema si risolve definendo la periferiche out, con il suo indirizzo, e la variabile v, posta all’indirizzo 400h, e contenente una sequenza di valori long terminata con uno 0.


Il programma azzera i registri da usare ed esegue una scansione di v e, componente per componente, con un offset di 4 trattandosi di longword, procede all’invio sulla periferica out. 


Per l’invio delle singole componenti di v alla periferica out, si usa la tecnica busy waiting. Man mano che si ispeziona ogni componente di v, il dato in essa contenuto viene consumato dal processo della periferica out. 
Registri usati:

R1: registro indice del vettore v
R2: elemento del vettore
Algoritmo

Inizio

   Azzeramento registri;

   Stampa sequenza;

Fine.

Stampa sequenza
Inizio
   R1 ( #v;

   Ripeti

       R2 ( v(R1)
       Se R2!=0

       Inizio

            Fintantochè dispositivo non pronto attendi;

            data buffer ( R2;

            Avvia dispositivo;

            R1 ( R1 + 4;

        Fine;
   Fino a che (R2=0);
Fine.

Codifica

;Programma che invia alla periferica p1 i valori 

;caricati in un vettore v posto all'indirizzo 500h.

;La sequenza di valori di v termina con 0h

;Autore: M.M.

;Data: 10/04/2004

;File: inout2.asm

ORG 400h

    v     DL  1,2,3,4,5,0

    out   EQU 10h       ;indirizzo disp. output

CODE

    JSR initregs           ;azzera registri

    JSR outvett            ;stampa elementi vettore

    JMP exit

;****** Inizializza registri usati ********

initregs:

    XORL R1, R1        ;indice vettore

    XORL R2, R2        ;elemento vettore

    RET

;****** Stampa elementi vettore ****************

outvett:

    MOVL #v, R1        ;indirizzo base di v

ciclo:

    MOVL (R1), R2

    CMPL #0, R2         ;fine vettore ?

    JZ enddata

    JSR dataout            ;invia dato alla periferica

    ADDL #4, R1        ;avanza al successivo

    JMP ciclo

enddata:

    RET

;****** Stampa elemento ************************

dataout:

outnotrd:

    JNR out, outnotrd

    OUTL R2, out

    START out

    RET

;******************************************

exit:

    HALT

END

Esercizio 12
Scrivere un programma PD32 che acquisisce 10 dati di tipo word da una periferica dev, posta all’indirizzo 17h e che, man mano, li memorizza in una zona di memoria iniziante dall’indirizzo 800h. 
Soluzione

Analisi
Il problema si risolve definendo la periferiche dev, con il suo indirizzo, e la variabile indbase equivalente all’indirizzo 800h, dal quale punto inizierà la memorizzazione dei dati in ingresso. 


Il programma azzera i registri da usare ed esegue un ciclo di lettura da dev dei singoli dati, memorizzandoli via via in locazioni successive con un offset di 2 trattandosi di longword.  


Per l’acquisizione dei singoli valori dalla periferica dev, si usa la tecnica busy waiting. Man mano che il dato viene prodotto dal processo della periferica dev, esso viene memorizzato opportunamente. 
Registri usati:

R0: numero dati da trasferire
R1: indirizzo base in memoria
R2: dato in ingresso
Algoritmo

Inizio

   Azzeramento registri;

   Caricamento sequenza;

Fine.

Caricamento sequenza
Inizio
   R0 ( ndati;

   R1 ( #indbase

   Fintantochè (R0!=0)

   Inizio

       Leggi dato;

      R0 ( R0-1;

   Fine;

Fine;

Leggi dato

Inizio

    Fintantochè dispositivo non pronto attendi;

    Avvia dispositivo;    

    Fintantochè dato non trasferito  attendi;
    R2 ( data buffer;
    (R1) ( R2;
    R1 ( R1 + 2;
Fine;
Codifica
;Programma che acquisisce 10 dati di tipo word da una 

;periferica dev di indirizzo 17h e li trasferisca in

;memoria a partire dall'indirizzo 800h.

;Utilizza la gestione di tipo Busy Waiting

;Autore: M.M.

;Data: 10/04/2004

;File: inout3.asm

ORG 400h

    ndati   DL 10             ;numero dati da trasferire

    dev     EQU 17h        ;indirizzo periferica

    indbase EQU 800h    ;indirizzo base

CODE

    JSR initregs               ;azzera registri

    JSR lettura

    JMP fine

;***** Inizializza registri *************************

initregs:

    XORL R0, R0            ;numero dati

    XORL R1, R1            ;indirizzo base

    XORL R2, R2            ;buffer per dati

    RET

;***** Acquisizione dati ****************************

lettura:

    MOVL ndati, R0

    MOVL #indbase, R1

next:

    CMPL #0, R0              ;fine dati
    JZ endread

    JSR indato                   ;acquisisci dato
    SUBL #1, R0

    JMP next        

endread:

    RET

;***** Lettura dato da periferica ******************

indato:

devnotrd:

    JNR dev, devnotrd       ;attesa dispositivo pronto

    START dev                 ;avvia produzione dato

datnotrd:

    JNR dev, datnotrd       ;attesa produzione dato

    INW dev, R2

    MOVW R2, (R1)+      ;memorizzazione dato
    RET

fine:

    HALT

END

Esercizio 13
Scrivere un programma PD32 che acquisisce 10 dati di tipo word da una periferica dev, posta all’indirizzo 17h e che, man mano, li memorizza in una zona di memoria iniziante dall’indirizzo 800h. A differenza dell’Esercizio 4, si richiede che la procedura di acquisizione dati venga scritta sotto forma di subroutine con parametri, in modo che sia generalizzabile rispetto al numero dei dati da acquisire, all’indirizzo di memoria in cui depositare i dati e al dispositivo.

Soluzione

Analisi

La soluzione è simile alla precedente, ma con la realizzazione della subroutine di acquisizione con i parametri richiesti. La gestione della periferica è busy waiting. 
Registri usati:

R0: numero dati da trasferire
R1: indirizzo base in memoria
R2: indirizzo periferica

R3: dato in ingresso

Algoritmo

Inizio

    Azzeramento registri;

   Preparazione parametri;

   Lettura dati;

Fine.

Preparazione parametri

Inizio

    R0 ( #ndati;
    R1 ( #indbase;
    R2 ( #dev;

Fine;

Lettura dati

Inizio

     Immissione parametri R0, R1, R2;

     Lettura sequenza;

     R7 ( R7 + 12;

Fine

Lettura sequenza

Inizio

     Salva R6;

     R6 ( R7;

     Alloca variabili locali R0, R1, R2, R3;

     R0 ( (R6)+16;   /* trasferimento parametri nei registri */

     R1 ( (R6)+12;

     R2 ( (R6)+8;

     Fintantochè (R0!=0)

     Inizio 

         Leggi dato;

         R0 ( R0-1;

      Fine

      Dealloca frame corrente  /* R3, R2, R1, R0, R6 */

Fine

Leggi dato

Inizio

     Fintantochè dispositivo non pronto attendi;

     Avvia dispositivo;    

     Fintantochè dato non trasferito  attendi;
     R3 ( data buffer;

     (R1) ( R3;
     R1 ( R1 + 2;
Fine

Codifica

;Programma che acquisisce 5 dati di tipo word da una 

;periferica dev di indirizzo 17h e li trasferisca in

;memoria a partire dall'indirizzo 800h.

;Utilizza la gestione di tipo Busy Waiting

;La subroutine di acquisizione è generalizzata e riceve

;come parametri:

; - R0: numero di dati da acquisire

; - R1: indirizzo base in cui depositare i dati

; - R2: indirizzo periferica

;Autore: M.M.

;Data: 10/04/2004

;File: inout4.asm

ORG 400h

    ndati   EQU 5       ;numero dati da trasferire

    dev     EQU 17h     ;indirizzo periferica

    indbase EQU 800h    ;indirizzo base

CODE

    JSR initregs        ;azzera registri

    JSR params          ;preparazione parametri 

    JSR reading

    JMP fine

;***** Inizializza registri *************************

initregs:

    XORL R0, R0         ;numero dati

    XORL R1, R1         ;indirizzo base

    XORL R2, R2         ;indirizzo periferica

    XORL R3, R3         ;buffer per dati

    RET

;***** Inizializza parametri ************************

params:

    MOVL #ndati, R0

    MOVL #indbase, R1

    MOVL #dev, R2

    RET

;***** Istanza con parametri ************************

reading:

    PUSH R0          ;preparazione parametri

    PUSH R1

    PUSH R2

    JSR datain 

    ADDL #12, R7     ;ripristina SP

    RET

;***** Lettura dati *********************************

datain:

    PUSH R6

    MOVL R7, R6

    PUSH R0          ;variabili locali

    PUSH R1

    PUSH R2

    PUSH R3

    MOVL 16(R6), R0  ;trasferimento valori dei parametri

    MOVL 12(R6), R1

    MOVL 8(R6), R2

next:

    CMPL #0, R0

    JZ endread

    JSR indato

    SUBL #1, R0

    JMP next        

endread:

    POP R3

    POP R2

    POP R1

    POP R0

    POP R6

    RET

;***** Lettura dato da periferica ******************

indato:

devnotrd:

    JNR R2, devnotrd    ;attesa dispositivo pronto

    START R2            ;avvia produzione dato

datnotrd:

    JNR R2, datnotrd    ;attesa produzione dato

    INW R2, R3

    MOVW R3, (R1)+

    RET

fine:

    HALT

END

Esercizio 14
   Dato un vettore v, scrivere un programma PD32 che, mediante due subroutine distinte, calcoli la somma degli elementi pari e quella dei dispari. Le somme ottenute vanno inviate in sequenza ad un dispositivo esterno out. 

Prima versione

;Dato un array di 100 byte memorizzato a partire dalla 

;locazione di memoria 400h, determinare:

;1) la somma dei numeri pari;

;2) la somma dei numeri dispari.

;Dopo aver effettuato tali somme, trasmetterle in sequenza

;ad un dispositivo di output avente indirizzo 30h.

;Le richieste 1) e 2) devono essere realizzate mediante 

;2 subroutine separate

;Autore: M.M.

;Data:05/05/2004

;File:sommapd.asm

ORG 400h

    v      DL 1,2,3,4,5,0    ;0 terminatore sequenza

    sommap DL 0

    sommad DL 0

    out    EQU 30h           ;device di uscita

CODE

    JSR initregs

    JSR sp

    JSR sd

    JSR outsp

    JSR outsd

    JMP fine

;***** Azzeramento registri *************************

initregs:

    XORL R0, R0         ;indirizzo base di v

    XORL R1, R1         ;accumulatore per somme

;    XORL R2, R2         ;accumulatore per sommad

;    XORL R4, R4         ;buffer di uscita

    XORL R5, R5         ;per testare parità indirizzo

RET

;***** Routine calcolo somma pari *******************

sp:

    XORL R1, R1

    MOVL #v, R0

    loopsp:

       MOVL (R0), R5

       CMPL #0, R5

       JZ endsp

       ASRL #1, R5

       JC dispari

       ADDL (R0), R1

       dispari:

          ADDL #4, R0 

          JMP loopsp       

    endsp:

    MOVL R1, sommap

RET        

;***** Routine calcolo somma dispari ****************

sd:

    XORL R1, R1

    MOVL #v, R0

    loopsd:

       MOVL (R0), R5

       CMPL #0, R5

       JZ endsd

       ASRL #1, R5

       JNC pari

       ADDL (R0), R1

       pari:

          ADDL #4, R0 

          JMP loopsd

    endsd:

    MOVL R1, sommad

RET        

;***** Routine emissione sommap *********************

outsp:

    waitsp:

       JNR out, waitsp

    OUTL sommap, out

    START out

RET

;***** Routine emissione sommad *********************

outsd:

    waitsd:

       JNR out, waitsd

    OUTL sommad, out

    START out

RET

fine:

    HALT

END

Seconda versione

;Dato un array di 100 byte memorizzato a partire dalla 

;locazione di memoria 400h, determinare:

;1) la somma dei numeri pari;

;2) la somma dei numeri dispari.

;Dopo aver effettuato tali somme, trasmetterle in sequenza

;ad un dispositivo di output avente indirizzo 30h.

;Le richieste 1) e 2) devono essere realizzate mediante 

;2 subroutine separate

;La routine di stampa è parametrica su R4 e la routine di

;somma è parametrica con flag R2

;Autore: M.M.

;Data:05/05/2004

;File:sommapd1.asm

ORG 400h

    v      DL 1,2,3,4,5,0    ;0 terminatore sequenza

    sommap DL 0

    sommad DL 0

    out    EQU 30h           ;device di uscita

CODE

    JSR initregs

    MOVB #0, R2         ;calcola somma pari

    JSR somma

    MOVB #1, R2         ;calcola somma dispari

    JSR somma

    JSR outdata

    JMP fine

;***** Azzeramento registri *************************

initregs:

    XORL R0, R0         ;indirizzo base di v

    XORL R1, R1         ;accumulatore per somme

    XORL R2, R2         ;flag: 0=somma pari

                        ;      1=somma dispari

    XORL R4, R4         ;buffer di uscita

    XORL R5, R5         ;per testare parità indirizzo

RET

;***** Routine calcolo somma pari *******************

somma:

    XORL R1, R1

    MOVL #v, R0

    loop:                ;ciclo esame vettore

       MOVL (R0), R5

       CMPL #0, R5

       JZ endloop        ;fine ciclo

       ASRL #1, R5       ;test pari/dispari

       JC odd

       JMP even

       odd:              ;R5 dispari

         CMPL #0, R2     ;se la somma richiesta...

         JZ go           ;...è pari vai a go

         ADDL (R0), R1   ;altrimenti somma elemento

       even:             ;R5 è pari

         CMPL #0, R2     ;se la somma richiesta...

         JNZ go          ;...è dispari vai a go

         ADDL (R0), R1   ;altrimenti somma elemento

       go: 

         ADDL #4, R0     ;avanza al successivo

         JMP loop

    endloop:

    CMPL #0, R2          ;tipo di somma richiesta

    JZ sp

       MOVL R1, sommad   ;risultato somma dispari

    JMP endsum

    sp:

       MOVL R1, sommap   ;risultato somma pari

  endsum:

RET        

;***** Routine emissione sommap *********************

outsp:

    waitsp:

       JNR out, waitsp

    OUTL sommap, out

    START out

RET

;***** Emissione risultati **************************

outdata:

    MOVL sommap, R4

    JSR outd

    MOVL sommad, R4

    JSR outd

RET

;***** Stampa dato **********************************

outd:

    waitsd:

       JNR out, waitsd

    OUTL R4, out

    START out

RET

fine:

    HALT

END

Esercizio 15

Supponendo di non avere a disposizione le interruzioni vettorizzate in PD32, realizzare un programma che, a ciclo continuo, testi le varie periferiche attive per individuare quale di esse abbia richiesto una interruzione. Il ciclo di polling avviene su 3 periferiche poste agli indirizzi 01h, 02h, 03h. I driver dei dispositivi, ossia le routine di gestione dell’interruzione 
Soluzione

Analisi

Gli indirizzi dei dispositivi vengono posti in un vettore devtbl[] in modo da reperirli con un ciclo. I rispettivi driver sono indicati con driver1, driver2 e driver3. Queste denominazioni sono associate alle rispettive label di programma e gli indirizzi ad esse associati sono posti in un secondo vettore drvtbl[]. Il programma principale, dopo le opportune inizializzazioni, esegue un ciclo continuo in attesa di un interrupt proveniente da una delle 3 periferiche. Una volta verificatosi l’interrupt, il programma, che per ipotesi non dispone delle interruzioni vettorizzate, deve eseguire un ciclo di polling per intercettare la periferica richiedente; una volta individuata la periferica, eseguirà il driver ad essa associato. 

Codifica
;Programma che intercetta, fra 3 dispositivi, quello che

;ha generato richiesta di interrupt, senza fare uso della

;IVT.

;Il programma usa due tabelle: una per gli indirizzi dei

;dispositivi, l'altra per gli indirizzi dei driver

;corrispondenti

;Autore: M.M.

;Data: 02/05/2004

;File: intrget.asm

ORG 400h

   dev1 EQU 01h       ;indirizzi fisici dei dispositivi

   dev2 EQU 02h       ;installati

   dev3 EQU 03h

   devtbl EQU 1200h   ;indirizzo tabella dispositivi

   drvtbl EQU 1000h   ;indirizzi driver dispositivi

   buffer DL 0        ;valore da trasferire alla periferica

   drvpllng EQU 1300h ;indirizzo driver centralizzato 

CODE

   JSR initregs       ;azzera registri

   JSR lddevtbl       ;caricamento devtbl

   JSR lddrvtbl       ;caricamento drvtbl

   JSR init           ;abilitazione interrupt,

                      ;inizializzazione registro indice R0

   JSR devsetim       ;abilitazione di tutti i dispositivi

   main:

      JMP main

   HALT

;***** Azzera registri *****************************

initregs:

   XORL R0, R0        ;indirizzo dispositivo corrente

   XORL R1, R1        ;indice devtbl

RET

;***** Caricamento tabella devices *****************

lddevtbl:

   MOVB #dev1, devtbl

   MOVB #dev2, devtbl+1

   MOVB #dev3, devtbl+2

   MOVB #0, devtbl+3         ;terminatore

RET

;***** Caricamento indirizzi drivers ***************

lddrvtbl:

   MOVL #driver1, drvtbl

   MOVL #driver2, drvtbl+4   ;offset di 4

   MOVL #driver3, drvtbl+8

   MOVL #0, drvtbl+12        ;terminatore

RET

;***** Inizializzazione ****************************

init:

   SETI                      ;abilitazione interrupt

   MOVL #devtbl, R1          ;indice di devtbl

RET

;***** Abilitazione dei dispositivi ****************

devsetim:

   setmloop:

      MOVB (R1), R0

      CMPB #0, R0

      JZ endloop

      SETIM R0

      ADDL #1, R1

      JMP setmloop

   endloop:

RET

;***** Ciclo di polling ****************************

polling:

   PUSH R0           ;salvataggio registri

   PUSH R1

   MOVL #devtbl, R1

   plloop:

      MOVB (R1), R0

      CMPB #0, R0

      JZ endpoll        ;fine esame dispositivi

      JR R0, founddev   ;trovato device richiedente

      ADDL #1, R1

   JMP plloop

   founddev:

      JSR execdrv       ;esecuzione driver relativo

endpoll:

   POP R1

   POP R0

RET

;***** Esecuzione driver richiesto ******************

execdrv:

   SUBL #devtbl, R1     ;calcolo indirizzo su drvtbl

   ASLL #2, R1

   ADDL #drvtbl, R1

   MOVL (R1), R0        ;R0=indirizzo del driver

   JSR (R0)

RET

;***** Driver delle periferiche *********************

;***** Driver centralizzato per la rilevazione *****

;gendrv:

   driver 0, drvpllng  ;
   MOVL #0, buffer

   JSR polling

RTI

driver1:

   MOVL buffer, R0

   ADDL #1, R0

   MOVL R0, buffer      ;per verificare che è stato

                        ;attivato il driver i-esimo

   OUTL R0, dev1        ;consumo del dato

   CLEAR dev1

RET

driver2:

MOVL buffer, R0

   ADDL #2, R0

   MOVL R0, buffer      ;per verificare che è stato

                        ;attivato il driver i-esimo

   OUTL R0, dev2        ;consumo del dato

   CLEAR dev2

RET

driver3:

MOVL buffer, R0

   ADDL #3, R0

   MOVL R0, buffer      ;per verificare che è stato

                        ;attivato il driver i-esimo

   OUTL R0, dev3        ;consumo del dato

   CLEAR dev3

RET

END      

Esercizio 16
Un vettore di n elementi word deve essere inviato ad una periferica di indirizzo 30h. L’invio dei dati deve avvenire soltanto al momento della ricezione di un segnale esterno proveniente da una periferica di indirizzo 20h.

Soluzione

Analisi
Il problema si risolve mediante l’uso delle interruzioni. Il dispositivo di ingresso, sigdev, che lancia il segnale è gestito mediante un apposito driver sigdrv. Dapprima si caricano i dati nel vettore; successivamente il programma principale si dispone in attesa di una interruzione richiesta da sigdev. Al momento dell’arrivo del segnale da questa periferica, parte il driver relativo che gestisce l’invio dei dati sulla periferica out, usando ad esempio, la tecnica busy waiting. La fine della trasmissione avviene quando un flag stop, inizialmente posto a 0, viene posto ad 1.
Il driver sigdrv, non essendoci particolari richieste del problema, può essere posto all’indirizzo 600h.
Registri usati:

R0: indirizzo del vettore

R1: contatore elementi

R2: elemento del vettore

R3: flag fine I/O
Algoritmo
Inizio

   Caricamento dati;

   Predisposizione sistema di interruzione;

   Ripeti

      Attesa interrupt;
   Fino a che stop=1;
   Disabilita interruzioni;

Fine.

Predisposizione sistema di interruzione

Inizio

   Abilita interruzioni;

   Abilita dispositivo;

   R3 ( 1;

Fine; 

Driver 2, sigdrv

   Salvataggio contesto;
   R0 ( #v;

   R1 ( n;
   Fintantochè (R1!=0)
      Inizio  

         Fintantochè out not ready attendi;

         Carica buffer;

         Avvia dispositivo;

         R1 ( R1-1;

      Fine
   Stop ( 1                    /* fine trasmissioni */
   Disabilita ulteriori interrupt;
   Ripristina il contesto

RTI

Codifica
;Programma di stampa di un vettore di n valori al momento

;della ricezione di un segnale di interruzione esterno

;Il registro R0 è usato per il segnale di interruzione

; R0=1 continua trasmissione

; R0=0 fine trasmissione

;Autore: M.M.

;Data: 10/05/2004

;File: signout.asm

ORG 500h

   v       EQU 400h          ;sequenza di valori
   n       DL 5

   out     EQU 30h           ;dispositivo di output

   sigdev  EQU 20h         ;indirizzo device esterno

                                        ;che invia il segnale

   sigdrv  EQU 600h        ;indirizzo driver

   stop    DB  0                 ;flag di fine I/O

CODE

   JSR loaddata                 ;carica vettore

   JSR inittx                      ;parametri trasmissione

repeat:

   CMPB stop, R3             ;se R3=0...

   JZ fine                            ;...fine trasmissione

   JMP repeat

;***** Inizializzazione parametri trasissione ******

inittx:

   SETI                                ;abilita interruzioni

   SETIM sigdev                 ;abilita sigdev a lanciare

                                            ;interruzioni

   MOVB #1, R3                 ;R3=1

RET

;***** Caricamento dati *****************************

loaddata:

   MOVL #v, R0

   MOVL n, R1

   MOVW #1, R2

   loop:                                 ;carica v con valori interi
      MOVW R2, (R0)+       ;progressivi
      ADDW #1, R2

      SUBL #1, R1

      JNZ loop

RET
;***** Driver che gestisce l'interrupt esterno ******

driver 2, sigdrv

   PUSH R0                 ;salvataggio contesto

   PUSH R1   

   MOVL #v, R0

   MOVL n, R1

   devnotrd:

      JNR out,devnotrd     ;ciclo di busy waiting

      OUTW (R0)+, out      ;carica buffer

      START out            ;avvia dispositivo

      SUBL #1, R1

      JNZ devnotrd         ;prepara altri dati

   MOVB #1, stop           ;fine trasmissioni

   CLRIM sigdev            ;impedisce altre interruzioni

   POP R1

   POP R0

RTI

fine:

   CLRI                              ;disabilita interrupt
   HALT                              

END      

Esercizio 17
Acquisire da input una serie di N valori da una periferica di indirizzo 12h e porli in un vettore con indirizzo base 400h. Utilizzare il sistema di interruzione.

Soluzione
Analisi
Dovendo usare il sistema di interruzioni, diciamo indrv il driver relativo al trattemento dell’interrupt relativo alla periferica in, suppponiamo che esso sia all’indirizzo 700h e che il suo IVN sia 2.
La soluzione migliore consiste nel progettare il driver in modo parametrico, ossia generale, per cui è bene usare i registri necessari come variabili locali al driver. Per fare questo, facciamo in modo di porre il numero di elementi N e l’indirizzo di v in variabili di memoria, rispettivamente cont e punt, caricarli in registri usati come variabili locali al driver e, dopo la loro modifica, riaggiornare il loro valore in memoria. Il ciclo continuo del programma principale usa cont come variabile di controllo.
Registri usati:

R0: contatore elementi

R1: puntatore a v

Algoritmo
Inizio

   Inizializza valori;

   Ripeti

        R1 ( cont;

        Attesa interrupt;

   Fino a che (R1=0);

Fine.

Inizializza valori

Inizio

    punt ( #v;

    cont ( #n;

    Abilita interruzioni;

   Abilita dispositivo;         /* L’abilitazione del device va fatta nel main, mentre la disabilitazione
                                               va posta nel driver (vedi commento seguente) */

Fine;

Driver 2, indrv
    Salva contesto;

    R0 ( punt;

    R1 ( cont;

   Carica buffer;

   Avvia dispositivo;

   R1 ( R1 -1;

   Se  (R1=0)

       Disabilita dispositivo;   /* è importante disabilitare qua il dispositivo, in quanto se



             per errore lo si disabilitasse nel main, al termine del ciclo





 di attesa dell’interrupt, il dispositivo potrebbe lanciare un nuovo

                                                interrupt e ripetere erroneamente la lettura  */

   Aggiorna cont e punt;

   Ripristina contesto;

RTI

Codifica    
;Programma che acquisisce N long word da un dispositivo

;di input, utilizzando un sistema di interruzione

;Autore: M.M.

;Data: 10/05/2004

;File: inter-in.asm

ORG 500h

   v  EQU 400h      ;punta alla sequenza elementi

   n  EQU 5            ;numero elementi

   in EQU 12h       ;indirizzo input device

   indrv EQU 700h   ;indirizzo driver input

   punt DL 0          ;puntatore a v

   cont DL 0           ;numero elementi

CODE

   JSR initrx           ;inizializza valori per ricezione

   loop:

      MOVL cont, R1

      CMPL #0, R1   ;se R1!=0...

      JNZ loop          ;...ciclo continuo. Il ciclo poteva essere anche controllato mediante…
                               ; …l’istruzione JIM   (loop: JIM in, loop)   

   JMP fine

;***** Inizializzazione valori per la trasmissione 

initrx:

   MOVL #v, punt    ;punt punta v

   SETI                    ;abilitazione interrupt

   SETIM in             ;abilitazione dispositivo

   MOVL #n, cont   

RET

;***** Driver gestione interrupt ********************

DRIVER 2, indrv     ;driver gestione periferica

   PUSH R0               ;salva il contesto 

   PUSH R1

   MOVL punt, R0    ;usa R0 locale

   MOVL cont, R1    ;usa R1 locale

   INL in, (R0)+        ;carica buffer

   START in

   SUBL #1, R1

   JZ endrx

   JMP next

endrx:

   CLRIM in

next:

   MOVL R0, punt    ;aggiorna punt e...

   MOVL R1, cont    ;...cont

   POP R1                  ;ripristina contesto

   POP R0

   RTI

fine:

   CLRI

   HALT

END

  Esercizio 18 

Si deve procedere all’acquisizione di N dati long word da un dispositivo di input. L’acquisizione di ciascun dato deve essere sincronizzata con un segnale di clock proveniente da un dispositivo esterno. Gestire l’acquisizione dei dati mediante sistema di interruzione.

Al termine dell’immissione dei dati si deve procedere al loro invio ad una periferica di output.

Soluzione

Analisi

Il problema non dà alcuna specifica per cui possiamo porre:

· Segnale di clock: dispositivo clkdev posto all’indirizzo 20h, driver relativo clkdrv posto all’indirizzo 700h e IVN pari a 2;

· Dispositivo di input indev posto all’indirizzo 30h;

· Dispositivo di output outdev posto all’indirizzo 40h.

Indichiamo con nval il numero di valori e con data l’indirizzo di inizio della sequenza di valori.

Il programma deve riconoscere un’interruzione proveniente da clkdev e acquisire il dato da indev (ad esempio con tecnica busy waiting). Dopo ogni lettura, il dispositivo clkdev deve essere posto a READY=0, per poter generare l’interrupt successivo. Al termine dell’acquisizione dei dati, occorre che IM del dispositivo clkdev sia resettato, in modo che esso non possa più lanciare interrrupt.

Al termine dell’immissione, i dati vengono emessi sul dispositivo di output, mediante tecnica busy waiting.

Algoritmo

Inizio

   Inizializza parametri di acquisizione; 

   Fintantoché cont !=0

       Attendi interruzione;

   Stampa valori;

Fine.

Inizializza parametri di acquisizione

Inizio

     R0 ( 0;

     R1 ( 0;

    cont ( #nval;

    punt ( #data;

    Abilita interruzioni;

    Abilita interruzioni clkdev;

Fine;

Driver 2, clkdrv

Inizio 

   Salva contesto;

   R0 ( punt;

   R1 ( cont;

   Lettura dato da indev;

   R0 ( R0 + 4;   

  punt ( R0;

   R1 ( R1-1;

  cont ( R1;

  Se cont=0

   Disabilita IM di clkdev ;  /* La CLRIM è opportuno venga eseguita all’interno del driver

                 


 perché se eseguita all’esterno, potrebbe dare luogo ad un 

  



 ulteriore interrupt, oltre agli N previsti                                 */

   Ripristina contesto;

   RTI

Fine;
Codifica

;Programma che acquisisce da input un vettore di N

;elementi word. L'acquisizione è sincronizzata con gli

;impulsi di un segnale di clock. Terminata l'immissione

;i dati vanno inviati ad una periferica di output.

;Autore: M.M.

;Data: 14/05/2004

;File: clkdrv.asm

ORG 400h

   clkdev   EQU 20h

   clkdrv   EQU 700h

   indev    EQU 30h

   outdev   EQU 40h

   data     EQU 800h

   nval     EQU 5

   cont     DL  0

   punt     DL  0

CODE

   JSR init

   loopext:

      MOVL cont, R1

      CMPL #0, R1

      JNZ loopext      ;ciclo continuo 

   JSR outdata

   JMP fine

;***** Inizializzazione parametri di acquisizione ****

init:

   XORL R0, R0

   XORL R1, R1

   MOVL #nval, cont   ;inizializza contatore

   MOVL #data, punt   ;inizializza puntatore al vettore

   SETI                         ;abilita interruzioni

   SETIM clkdev          ;abilita interruzioni device

RET

;***** Stampa dati ***********************************

outdata:

   PUSH R0                  ;salvataggio contesto

   PUSH R1

   MOVL #nval, R1

   MOVL #data, R0

   outloop:

      JNR outdev, outloop

      OUTL (R0)+, outdev

      START outdev

      SUBL #1, R1

      JNZ outloop

   POP R1                        ;ripristino contesto

   POP R0

RET

;***** Driver clock device ***************************

DRIVER 2, clkdrv

   PUSH R0                      ;salva contesto

   PUSH R1

   MOVL punt, R0

   MOVL cont, R1

   inloop:

      JNR indev, inloop       ;busy waiting

   START indev  

   INL indev, (R0)+

   MOVL R0, punt

   SUBL #1, R1

   MOVL R1, cont

   JZ endrx

   CLEAR clkdev                ;attende la prossima interruzione

   JMP enddata

   endrx:

      CLRIM clkdev              ;termine dell'immissione

   enddata:

      POP R1

      POP R0

RTI  

fine:

   HALT

   END

